## Uncertainty and Error Propagation

Propague

```
Propague(str#) :=
  s# := strsplit(substr(str#, 5, strlen(str#) − 5), ",")
  [f# x# ans#] := [s#_1 s#_[2..rows(s#)] 0]
  ξ# := str2num(x#)
  for k# ∈ [1..length(x#)]
    str2num(concat("Clear(", x#_k#, ")"))
    d# := str2num(concat("diff(", f#, ",", x#_k#, ")"))
    str2num(concat(x#_k#, ":", num2str(ξ#_k#)))
    if substr(str#, 1, 4) = "Err("
      ans# := ans# + |d# · str2num(concat(#Propague_1, x#_k#))|
    else if substr(str#, 1, 4) = "Unc("
      ans# := ans# + d#^2 · (str2num(concat(#Propague_2, x#_k#)))^2
    else
      error("Not implemented.")
  if substr(str#, 1, 4) = "Err("
    ans#
  else
    √ans#
```

ToDo: Round according correct figures.

## Examples

#Propague := [ "Δ" "s." ]

Specifies the symbols for the error and the standard deviation

Propague("Err(f,a,b,...)")

Returns $\Delta f = \left|\frac{\partial f}{\partial a}\cdot \Delta a\right| + \left|\frac{\partial f}{\partial b}\cdot \Delta b\right| + \dots$

Propague("Unc(f,a,b,...)")

Returns $s_f = \sqrt{\left(\frac{\partial f}{\partial a}\right)^2 \cdot s_a{}^2 + \left(\frac{\partial f}{\partial b}\right)^2 \cdot s_b{}^2 + \dots}$

**Error Propagation**

$[R_1\ \Delta R_1] := [350\ \Omega\ 5\ \Omega]$ $[R_2\ \Delta R_2] := [25\ k\Omega\ 10\ \Omega]$ $[I\ \Delta I] := [35\ mA\ 1\ mA]$

$V := \left|I \cdot \left(\frac{1}{R_1} + \frac{1}{R_2}\right)^{-1}\right.$ $V = 12.0809\ V$

Use line() for dealy the evaluation

$\Delta V :=$ Propague("Err(V,R.1,R.2,I)") $= 0.5154\ V$ $V \pm \Delta V = \begin{cases} 12.5963 \\ 11.5654 \end{cases} V$

**Sum & Product Uncertainty**

$[a\ s_a] := [3\ m\ 2\ cm]$ $[b\ s_b] := [5\ m\ 2\ cm]$

$L := |2 \cdot (a + b)$ $L = 16\ m$

$s_L :=$ Propague("Unc(L,a,b)") $= 5.6569\ cm$ $L \pm s_L = \begin{cases} 16.0566 \\ 15.9434 \end{cases} m$

$S := |a \cdot b$ $S = 15\ m^2$

$s_S := Propague("Unc(S,a,b)") = 1166.1904\ cm^2$ $\quad S \pm s_S = \begin{cases} 15.1166 \\ 14.8834 \end{cases} m^2$

Check $\quad 2 \cdot \sqrt{s_a^2 + s_b^2} = 5.6569\ cm$ $\quad S \cdot \sqrt{\left(\frac{s_a}{a}\right)^2 + \left(\frac{s_b}{b}\right)^2} = 1166.1904\ cm^2$

**Power Uncertainty**

$[x\ s_x] := [4\ m\ 2\ cm]$

$f := |x^2$ $\quad f = 16\ m^2$

$s_f := Propague("Unc(f,x)") = 1600\ cm^2$ $\quad f \pm s_f = \begin{cases} 16.16 \\ 15.84 \end{cases} m^2$

Notice that it is different to the product of x by x

$y := x$ $\quad s_y := s_x$ $\quad h := |x \cdot y$ $\quad h = 16\ m^2$

$s_h := Propague("Unc(h,x,y)") = 1131.3708\ cm^2$

**Uncertainty Example**

$[a\ s_a] := [3.02\ 0.01]$ $\quad [b\ s_b] := [4.00\ 0.05]$

$[c\ s_c] := [-1.25\ 0.02]$ $\quad [\tau\ s_\tau] := [0.80\ 0.04]$

$k := |(a+b) \cdot e^{c \cdot \tau}$ $\quad k = 2.5825$

$s_k := Propague("Unc(k,a,b,c,\tau)") = 0.1369$ $\quad k \pm s_k = \begin{cases} 2.7194 \\ 2.4456 \end{cases}$

Check

$v_1 := a + b$ $\quad s_1 := \sqrt{s_a^2 + s_b^2}$

$v_2 := c \cdot \tau$ $\quad s_2 := v_2 \cdot \sqrt{\left(\frac{s_c}{c}\right)^2 + \left(\frac{s_\tau}{\tau}\right)^2}$

$v_3 := e^{c \cdot \tau}$ $\quad s_3 := e^{v_2} \cdot s_2$

$v_4 := v_1 \cdot v_3$ $\quad s_4 := v_4 \cdot \sqrt{\left(\frac{s_1}{v_1}\right)^2 + \left(\frac{s_3}{v_3}\right)^2} = 0.1369$

---

*Alvaro*